> Laplace anti-tranform for complex and multiple roots

Error, missing operator or `;`

```
> with(inttrans) ;
```

$$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable] \quad (1)$$

Simple complex root

```
> R := (s+1)/(s^2+s+1) ;
```

$$R := \frac{s+1}{s^2+s+1} \quad (1.1)$$

```
> invlaplace(R,s,t);
```

$$\frac{1}{3} e^{-\frac{1}{2}t} \left(3 \cos\left(\frac{1}{2}\sqrt{3}\, t\right) + \sqrt{3} \sin\left(\frac{1}{2}\sqrt{3}\, t\right)\right) \quad (1.2)$$

First of all transform R as a sum of simple fraction

```
> convert( R, fullparfrac): pf := allvalues(%) ;
```

$$pf := \frac{-\frac{1}{6} I\sqrt{3} + \frac{1}{2}}{s - \frac{1}{2} I\sqrt{3} + \frac{1}{2}} + \frac{\frac{1}{2} + \frac{1}{6} I\sqrt{3}}{s + \frac{1}{2} + \frac{1}{2} I\sqrt{3}} \quad (1.3)$$

Extract the roots of denominator of R=P/Q

```
> roots_of_Q := solve( denom(R), {s} ) ;
```

$$roots\_of\_Q := \left\{s = \frac{1}{2} I\sqrt{3} - \frac{1}{2}\right\}, \left\{s = -\frac{1}{2} - \frac{1}{2} I\sqrt{3}\right\} \quad (1.4)$$

Extract alpha and p for alpha/(s-p) + conjugate(alpha)/(s-conjugate(p))

```
> alpha := 1/2-(1/6*I)*sqrt(3);
  p     := -(1/2-(1/2*I)*sqrt(3));
```

$$\alpha := -\frac{1}{6} I\sqrt{3} + \frac{1}{2}$$

$$p := \frac{1}{2} I\sqrt{3} - \frac{1}{2} \quad (1.5)$$

Rebuild the polynomial P and Q

```
> Q := s^2-2*Re(p)*s+abs(p)^2 ;
  P := s*2*Re(alpha) -2*Re(alpha*conjugate(p)) ;
```

$$Q := s^2 + s + 1$$

$$P := s + 1 \quad (1.6)$$

Complete the square for Q

```
> a := -1/2 ;
```

(1.7)

$$a := -\frac{1}{2} \tag{1.7}$$

```
> omega := sqrt(simplify(Q-(s-a)^2)) ;
```

$$\omega := \frac{1}{2}\sqrt{3} \tag{1.8}$$

Using table with a = -1/2 and omega=1/2*sqrt(3) --> (s+1)/((s-a)^2+omega^2)
(s+1)/((s-a)^2+omega^2) = (s-a)/((s-a)^2+omega^2)+ (a+1)/((s-a)^2+omega^2)*(omega/omega)
=(s-a)/((s-a)^2+omega^2)+(a+1)/omega * omega/((s-a)^2+omega^2);

using the Table

```
> invlaplace( (s-A)/((s-A)^2+Omega^2), s, t ) ;
  invlaplace( Omega/((s-A)^2+Omega^2), s, t ) ;
```

$$e^{A t}\cos(\Omega t)$$

$$e^{A t}\sin(\Omega t) \tag{1.9}$$

```
> sol := exp(a*t)*cos(omega*t) + (a+1)/omega * exp(a*t)*sin(omega*
  t) ;
```

$$sol := e^{-\frac{1}{2}t}\cos\left(\frac{1}{2}\sqrt{3}\,t\right) + \frac{1}{3}\sqrt{3}\,e^{-\frac{1}{2}t}\sin\left(\frac{1}{2}\sqrt{3}\,t\right) \tag{1.10}$$

```
> simplify( sol - invlaplace(R,s,t) ) ;
```

$$0 \tag{1.11}$$

Double complex root

```
> R := (s+1)/(s^2+s+1)^2 ;
```

$$R := \frac{s+1}{\left(s^2+s+1\right)^2} \tag{2.1}$$

```
> invlaplace(R,s,t);
```

$$\frac{1}{9}e^{-\frac{1}{2}t}\left(-3\,t\cos\left(\frac{1}{2}\sqrt{3}\,t\right) + \sin\left(\frac{1}{2}\sqrt{3}\,t\right)\sqrt{3}\,(2+3\,t)\right) \tag{2.2}$$

First of all transform R as a sum of simple fraction

```
> convert( R, fullparfrac): pf := allvalues(%) ;
```

$$pf := \frac{-\frac{1}{6}I\sqrt{3}-\frac{1}{6}}{\left(s-\frac{1}{2}I\sqrt{3}+\frac{1}{2}\right)^2} + \frac{-\frac{1}{6}+\frac{1}{6}I\sqrt{3}}{\left(s+\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)^2} - \frac{\frac{1}{9}I\sqrt{3}}{s-\frac{1}{2}I\sqrt{3}+\frac{1}{2}} + \frac{\frac{1}{9}I\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}I\sqrt{3}} \tag{2.3}$$

```
> int(op(1,pf),s);
  int(op(2,pf),s);
```

$$-\frac{-\frac{1}{6}\,\mathrm{I}\,\sqrt{3}-\frac{1}{6}}{s-\frac{1}{2}\,\mathrm{I}\,\sqrt{3}+\frac{1}{2}}-\frac{-\frac{1}{6}+\frac{1}{6}\,\mathrm{I}\,\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\,\sqrt{3}} \tag{2.4}$$

Tranform higher powers as deriovative of lower powers.

```
> pf_new := Diff( int(op(1,pf)+op(2,pf),s), s) + op(3,pf)+ op(4,pf)
;
```

$$pf\_new := \frac{d}{ds}\left(-\frac{-\frac{1}{6}\,\mathrm{I}\,\sqrt{3}-\frac{1}{6}}{s-\frac{1}{2}\,\mathrm{I}\,\sqrt{3}+\frac{1}{2}}-\frac{-\frac{1}{6}+\frac{1}{6}\,\mathrm{I}\,\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\,\sqrt{3}}\right)-\frac{\frac{1}{9}\,\mathrm{I}\,\sqrt{3}}{s-\frac{1}{2}\,\mathrm{I}\,\sqrt{3}+\frac{1}{2}}+\frac{\frac{1}{9}\,\mathrm{I}\,\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\,\sqrt{3}} \tag{2.5}$$

```
> R1 := op(2,pf_new)+op(3,pf_new); R1_new := simplify(numer(R1)
/expand(denom(R1))) ;
```

$$R1 := -\frac{\frac{1}{9}\,\mathrm{I}\,\sqrt{3}}{s-\frac{1}{2}\,\mathrm{I}\,\sqrt{3}+\frac{1}{2}}+\frac{\frac{1}{9}\,\mathrm{I}\,\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\,\sqrt{3}}$$

$$R1\_new := \frac{1}{3\,(s^2+s+1)} \tag{2.6}$$

```
> R2 := op(1,op(1,pf_new)); R2_new := simplify(numer(R2)/expand
(denom(R2))) ;
```

$$R2 := -\frac{-\frac{1}{6}\,\mathrm{I}\,\sqrt{3}-\frac{1}{6}}{s-\frac{1}{2}\,\mathrm{I}\,\sqrt{3}+\frac{1}{2}}-\frac{-\frac{1}{6}+\frac{1}{6}\,\mathrm{I}\,\sqrt{3}}{s+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\,\sqrt{3}}$$

$$R2\_new := \frac{1}{3}\,\frac{-1+s}{s^2+s+1} \tag{2.7}$$

```
> R_new := Diff(R2_new,s) + R1_new ;
```

$$R\_new := \frac{d}{ds}\left(\frac{1}{3}\,\frac{-1+s}{s^2+s+1}\right)+\frac{1}{3\,(s^2+s+1)} \tag{2.8}$$

```
> simplify(eval(subs(Diff=diff,R_new)-R))  ;
```

$$0 \tag{2.9}$$

```
> Q := s^2+s+1 ;
```

$$Q := s^2+s+1 \tag{2.10}$$

Complete the square for Q

```
> a := -1/2 ;
```

$$a := -\frac{1}{2} \tag{2.11}$$

```
> omega := sqrt(simplify(Q-(s-a)^2)) ;
```

$$\omega := \frac{1}{2}\sqrt{3} \tag{2.12}$$

Using table with a = -1/2 and omega=1/2*sqrt(3) --> (s+1)/((s-a)^2+omega^2)
(s-1)/((s-a)^2+omega^2) = (s-a)/((s-a)^2+omega^2)+ (a-1)/((s-a)^2+omega^2)*(omega/omega)
=(s-a)/((s-a)^2+omega^2)+(a-1)/omega * omega/((s-a)^2+omega^2);

using the Table

```
> invlaplace( (s-A)/((s-A)^2+Omega^2), s, t ) ;
  invlaplace( Omega/((s-A)^2+Omega^2), s, t ) ;
```

$$e^{A t}\cos(\Omega t)$$

$$e^{A t}\sin(\Omega t) \tag{2.13}$$

```
> sol := -t*(invlaplace((1/3)*(s-1)/(s^2+s+1),s,t))+
         invlaplace(1/(3*(s^2+s+1)),s,t) ;
```

$$sol := -\frac{1}{3}\, t\, e^{-\frac{1}{2} t}\left(\cos\left(\frac{1}{2}\sqrt{3}\, t\right) - \sqrt{3}\sin\left(\frac{1}{2}\sqrt{3}\, t\right)\right) + \frac{2}{9}\sqrt{3}\, e^{-\frac{1}{2} t}\sin\left(\frac{1}{2}\sqrt{3}\, t\right) \tag{2.14}$$

```
> simplify( sol - invlaplace(R,s,t) ) ;
```

$$0 \tag{2.15}$$

A simple problem with complex root

```
> restart:
> with(inttrans):
> ODE1 := diff(x(t),t)+diff(y(t),t)=sin(t);
  ODE2 := diff(x(t),t)-diff(y(t),t)=1;
  BC := x(0)=0, y(Pi)=0 ;
```

$$ODE1 := \frac{d}{dt}x(t) + \frac{d}{dt}y(t) = \sin(t)$$

$$ODE2 := \frac{d}{dt}x(t) - \left(\frac{d}{dt}y(t)\right) = 1$$

$$BC := x(0) = 0, y(\pi) = 0 \tag{3.1}$$

Use Laplace to transform into an algebraic problem

```
> ODE1s := laplace ( ODE1, t, s) ;
  ODE2s := laplace ( ODE2, t, s) ;
```

$$ODE1s := s\, laplace(x(t), t, s) - x(0) + s\, laplace(y(t), t, s) - y(0) = \frac{1}{s^2+1}$$

(3.2)

$$ODE2s := s\,laplace(x(t), t, s) - x(0) - s\,laplace(y(t), t, s) + y(0) = \frac{1}{s} \quad (3.2)$$

Apply BC

```
> ODE1sbc := subs(BC,ODE1s) ;
  ODE2sbc := subs(BC,ODE2s) ;
```

$$ODE1sbc := s\,laplace(x(t), t, s) + s\,laplace(y(t), t, s) - y(0) = \frac{1}{s^2+1}$$

$$ODE2sbc := s\,laplace(x(t), t, s) - s\,laplace(y(t), t, s) + y(0) = \frac{1}{s} \quad (3.3)$$

Solve the transformed system

```
> SOL := solve( {ODE1sbc,ODE2sbc},{laplace(x(t), t, s),laplace(y
  (t), t, s)}) ;
```

$$SOL := \left\{ laplace(x(t), t, s) = \frac{1}{2}\,\frac{s+s^2+1}{s^2\,(s^2+1)},\ laplace(y(t), t, s) = \frac{1}{2}\,\frac{2\,y(0)\,s^3+2\,y(0)\,s+s-s^2-1}{s^2\,(s^2+1)} \right\} \quad (3.4)$$

```
> XSOL_by_s := convert( subs( SOL, laplace(x(t), t, s)), parfrac) ;
  YSOL_by_s := convert( subs( SOL, laplace(y(t), t, s)), parfrac,
  s) ;
```

$$XSOL\_by\_s := -\frac{1}{2}\,\frac{s}{s^2+1} + \frac{1}{2\,s^2} + \frac{1}{2\,s}$$

$$YSOL\_by\_s := -\frac{1}{2}\,\frac{s}{s^2+1} - \frac{1}{2\,s^2} + \frac{1}{2}\,\frac{2\,y(0)+1}{s} \quad (3.5)$$

Use table to invert (here we use invlaplace)

```
> X := invlaplace( XSOL_by_s, s, t ) ;
  Y := invlaplace( YSOL_by_s, s, t ) ;
```

$$X := -\frac{1}{2}\cos(t) + \frac{1}{2}\,t + \frac{1}{2}$$

$$Y := -\frac{1}{2}\cos(t) - \frac{1}{2}\,t + y(0) + \frac{1}{2} \quad (3.6)$$

Find BC for determine y(0)

```
> SOL_for_y0 := solve( subs( t=Pi, Y ), {y(0)} ) ;
```

$$SOL\_for\_y0 := \left\{ y(0) = \frac{1}{2}\cos(\pi) + \frac{1}{2}\,\pi - \frac{1}{2} \right\} \quad (3.7)$$

```
> eval(subs( SOL_for_y0, [X,Y] ));
```

$$\left[-\frac{1}{2}\cos(t) + \frac{1}{2}\,t + \frac{1}{2},\ -\frac{1}{2}\cos(t) - \frac{1}{2}\,t - \frac{1}{2} + \frac{1}{2}\,\pi \right] \quad (3.8)$$